ECONOMIZATION IN NUMBER OF SURFACE PANEL MOLDS

FRED SCHWAB
February 15, 1990

I. INTRODUCTION

In GBT Memo No. 28 (*Geometry of the Primary Surface of the GBT*) I gave a rough calculation of the least number of precision molds that might be required for fabrication of panels for the primary reflecting surface of the Green Bank Telescope. Here I consider that question in further detail.

Letters received in mid-January from Jerry Nelson and Sebastian von Hoerner also address this general subject (GBT Memoranda Nos. 31 and 32). I believe that my results are in satisfactory agreement with theirs.

II. METHODOLOGY

For convenience' sake, in Memo 28 (where I began by addressing the possible use of spherical panels), I took as the nominal figure of each surface panel not an actual portion of the design paraboloid, but rather a figure with a circular profile at every angle α through the center of the panel, with the appropriate sectional curvature ($\kappa_1 \cos^2 \alpha + \kappa_2 \sin^2 \alpha$) at each of these angles. However, in this memorandum I will assume that the panels are fabricated to match the actual design paraboloid, since that case is, in fact, easy enough to deal with computationally. Since paraboloidal panel shapes are probably no problem to manufacture, I believe it is sensible to discard that earlier assumption (and this eliminates an additional, minor source of error).

Also, in Memo 28 I commented that the r.m.s. surface errors quoted there could be reduced by re-distributing the errors equally behind, and in front of, the design paraboloid—i.e., by offsetting each panel appropriately, in the direction of a surface normal central to the region of interest. The results presented below include such an offset, which typically reduces the r.m.s. error to about 58% of the r.m.s. achieved without any offset.

Tangent Plane Coordinates. Consider a point $\mathbf{x}$ on the surface, with Cartesian coordinates $\mathbf{x} = (x, y, \frac{x^2+y^2}{4c}) = (r\cos\varphi, r\sin\varphi, \frac{r^2}{4c})$ (where, as before, c is the focal length, nominally 60 meters, $r = \sqrt{x^2+y^2}$, and $\varphi = \tan^{-1}\frac{y}{x}$). Then the plane tangent to the surface at $\mathbf{x}$ is given by

$$\mathbf{X} = \mathbf{x} + \xi \frac{\frac{\partial \mathbf{x}}{\partial r}}{\left|\frac{\partial \mathbf{x}}{\partial r}\right|} + \eta \frac{\frac{\partial \mathbf{x}}{\partial \varphi}}{\left|\frac{\partial \mathbf{x}}{\partial \varphi}\right|}, \tag{1}$$

where ξ and η are allowed to vary. The normalization in Equation 1 is such that a unit step in the tangent plane coordinates (ξ, η) is a unit step in the units of c (e.g., meters). Equation 1 can be written explicitly, but more cumbersomely, as

$$\mathbf{X} = \left(\left(1 + \frac{2c\xi}{r\sqrt{r^2 + 4c^2}} \right) x - \frac{\eta}{r} y,\ \frac{\eta}{r} x + \left(1 + \frac{2c\xi}{r\sqrt{r^2 + 4c^2}} \right) y,\ \frac{r^2}{4c} + \frac{\xi r}{\sqrt{r^2 + 4c^2}} \right). \tag{2}$$

From a given point in the tangent plane, what is the distance to the paraboloid, measured along a line perpendicular to that plane? The unit normal at $\mathbf{x}$ is given by $\mathbf{N} = \frac{(-x,-y,2c)}{\sqrt{r^2+4c^2}}$, and a plane parallel to the tangent plane, and a distance t away from it, has equation

$$\mathbf{X}_t = \mathbf{x} + \xi \frac{\frac{\partial \mathbf{x}}{\partial r}}{\left|\frac{\partial \mathbf{x}}{\partial r}\right|} + \eta \frac{\frac{\partial \mathbf{x}}{\partial \varphi}}{\left|\frac{\partial \mathbf{x}}{\partial \varphi}\right|} + t\mathbf{N}. \tag{3}$$

If a point in the plane $\mathbf{X}_t$ is given by (x_t, y_t, z_t), and if that point lies on the paraboloid, then x_t, y_t, and z_t must satisfy $x_t^2 + y_t^2 = 4cz_t$. Written out in full, this is just a quadratic equation in t; its two roots are

$$t = \frac{a^3 + 2c\xi r \pm a\sqrt{4ac\xi r + (4c^2 - a^2)\eta^2 + a^4}}{r^2}, \tag{4}$$

where $a \equiv \sqrt{r^2 + 4c^2}$. Choosing the 'minus' sign yields the desired root.

Let us regard t, of Equation 4, as a function of r, ξ, and η. What is the mismatch between a panel designed for a location r_0, and a patch of the design paraboloid centered at a different radius r? The r.m.s. difference σ, calculated by integrating over a region A in (ξ, η), and allowing a fixed offset o, is given by

$$\sigma^2 = \frac{\iint_A \left(t(r,\xi,\eta) - (t(r_0,\xi,\eta) + o)\right)^2 d\xi\, d\eta}{\iint_A d\xi\, d\eta}. \tag{5}$$

A Fortran program to numerically evaluate σ, over rectangular regions A, is given in the Appendix. The offset o is set equal in the program to one-half the average of the minimum and maximum errors that would occur on the boundary of A if no offset were included.

This methodology is similar to that used by von Hoerner, except for inclusion of a constant offset, and except that Equation 4 is exact, whereas an approximation is used in Memo 32.[1]

III. Results

Numerical results are summarized in Tables 1, 2, and 3. Table 1, based on twelve molds, is a computation for a circular region A, of diameter 2.5 meters. This calculation is not of great interest, because it would be most relevant to the case of hexagonal panel shapes—an idea that I believe we have discarded. I have included

[1] My Equation 4 corresponds to Equation 4 of Memo 32, upon substituting $U = \xi$, $V = \eta$, $F = c$, and $\alpha = \tan^{-1} \frac{r}{2c}$ to convert to the notation of Memo 32. For the panel sizes we are considering, von Hoerner's approximation would be entirely adequate.

this table to provide greater continuity with Memo 28. The similar case shown in Memo 28 had r.m.s. errors which were typically near 70 μm. Here the r.m.s. errors are generally below 30 μm; the difference is due primarily to my having included an offset (see Eq. 5). The remainder of the difference, at small radii, is due to the use of one more mold than in Table 2 of the earlier memo; at large radii, the remainder of the difference is due to the fact that here I am integrating over circular regions in the tangent planes, rather than circular regions in the x-y plane, and to the fact that distances here are measured parallel to a central surface normal, rather than along individual normals.

In the fifth column of each table is an overall r.m.s. error, calculated by assuming an additional 75 μm (r.m.s.) of random error in panel manufacture. The sixth column shows the maximum component of systematic error, which always occurs somewhere on the perimeter of the panel.

Table 2 is a calculation for the case of thirteen molds, for rectangular panels 2.5 m $\times$ 2 m in size. Here, the r.m.s. errors are all below 27 μm, which is excellent. The problem is that the maximum errors are quite large, $\sim$ 100 μm in several instances, and these errors occur along the panel boundaries.

Ridges of systematic error would occur along meridianal panel seams (φ = constant), with jump discontinuities at the ring boundaries. Everywhere along certain parallels (i.e., between panel rings, with r = constant) there would exist maxima in the error pattern, as well as discontinuities all along these lines. This situation is illustrated in Figure 1.

Whether such patterns of systematic error would be tolerable should perhaps be the subject of studies in the electromagnetics of the Green Bank Telescope design. I wouldn't know how to address this question, except by numerical simulation.

Table 3 shows the effect of increasing the number of molds from thirteen to twenty. Here, the maximum errors are approximately halved, to $\sim$ 50 μm in absolute value. I did not attempt to optimize the mold usage for this case; obviously if the errors at the larger radii are tolerable, then even somewhat fewer molds would suffice at the smallest radii.

IV. Conclusions

The results I have presented are enticing, because, translated into tooling costs, they could represent savings as great as 50% or 67.5% of the original cost estimate for panel molds (based on forty molds)—and panel molds will be expensive.[2] Electromagnetic studies should address the deleterious effect of such systematic primary surface errors as I have described. If it turns out that some level of this error would be tolerable, then, once limits are established, we should be able to quickly and easily optimize, for greatest economy, the panel geometry and the mold requirements.

If these types of error are not acceptable, then perhaps the formal design specifications should be made more specific than they are at present; e.g., the maximum allowable absolute error should be specified in addition to the maximum r.m.s. error.

[2] Lee King pointed out at the Feb. 12 meeting of the design group that details of panel manufacture determine whether these savings would indeed be realizable. It might be that molds will need to be uniquely fabricated for each choice of panel boundary, as well as panel surface shape.

Table 1. Panel Mold Utilization (12 Molds, Panel Diameter = 2.5 m)

Panel Location, r (meters)	Design Radius (meters)	Offset (microns)	R.m.s., σ (microns)	$\sqrt{(75\mu\text{m})^2 + \sigma^2}$ (microns)	Max. |Error| (microns)
5.25	15.25	46.7	32.1	81.6	94.5
7.75	15.25	39.1	26.9	79.7	78.9
10.25	15.25	28.7	19.8	77.6	57.8
12.75	15.25	15.7	10.8	75.8	31.5
15.25	15.25	0.0	0.0	75.0	0.0
17.75	15.25	−18.2	12.6	76.0	−36.4
20.25	15.25	−38.8	26.9	79.7	−77.5
22.75	25.25	25.1	17.4	77.0	49.6
25.25	25.25	0.0	0.0	75.0	0.0
27.75	25.25	−27.1	18.8	77.3	−53.5
30.25	32.75	30.8	21.3	78.0	60.3
32.75	32.75	0.0	0.0	75.0	0.0
35.25	32.75	−32.5	22.4	78.3	−63.2
37.75	40.25	35.3	24.3	78.8	68.0
40.25	40.25	0.0	0.0	75.0	0.0
42.75	40.25	−36.5	25.0	79.1	−69.9
45.25	47.75	38.5	26.2	79.5	72.8
47.75	47.75	0.0	0.0	75.0	0.0
50.25	47.75	−39.3	26.7	79.6	−73.8
52.75	55.25	40.4	27.3	79.8	75.0
55.25	55.25	0.0	0.0	75.0	0.0
57.75	55.25	−40.9	27.5	79.9	−75.3
60.25	62.75	41.4	27.6	79.9	75.2
62.75	62.75	0.0	0.0	75.0	0.0
65.25	62.75	−41.5	27.5	79.9	−74.8
67.75	70.25	41.4	27.2	79.8	73.4
70.25	70.25	0.0	0.0	75.0	0.0
72.75	70.25	−41.2	26.9	79.7	−72.5
75.25	77.75	40.7	26.3	79.5	70.3
77.75	77.75	0.0	0.0	75.0	0.0
80.25	77.75	−40.3	26.0	79.4	−69.0
82.75	85.25	39.4	25.1	79.1	66.1
85.25	85.25	0.0	0.0	75.0	0.0
87.75	85.25	−38.9	24.7	79.0	−64.5
90.25	92.75	37.8	23.7	78.7	61.3
92.75	92.75	0.0	0.0	75.0	0.0
95.25	92.75	−37.2	23.2	78.5	−59.6
97.75	100.25	35.9	22.2	78.2	56.2
100.25	100.25	0.0	0.0	75.0	0.0
102.75	100.25	−35.2	21.7	78.1	−54.5

Table 2. Panel Mold Utilization (13 Molds, 2.5 m $\times$ 2 m Panels)

Panel Location, r (meters)	Design Radius (meters)	Offset (microns)	R.m.s., σ (microns)	$\sqrt{(75\mu m)^2 + \sigma^2}$ (microns)	Max. |Error| (microns)
5.25	10.25	19.7	11.6	75.9	47.9
7.75	10.25	11.3	6.7	75.3	27.4
10.25	10.25	0.0	0.0	75.0	0.0
12.75	10.25	−14.2	8.4	75.5	−34.3
15.25	17.75	19.6	11.7	75.9	47.3
17.75	17.75	0.0	0.0	75.0	0.0
20.25	17.75	−22.2	13.3	76.2	−53.4
22.75	25.25	27.0	16.3	76.7	64.8
25.25	25.25	0.0	0.0	75.0	0.0
27.75	25.25	−29.2	17.6	77.0	−69.9
30.25	32.75	33.2	20.0	77.6	79.1
32.75	32.75	0.0	0.0	75.0	0.0
35.25	32.75	−35.0	21.1	77.9	−83.1
37.75	40.25	38.1	22.9	78.4	90.0
40.25	40.25	0.0	0.0	75.0	0.0
42.75	40.25	−39.4	23.7	78.7	−92.8
45.25	47.75	41.5	25.0	79.1	97.4
47.75	47.75	0.0	0.0	75.0	0.0
50.25	47.75	−42.4	25.5	79.2	−99.1
52.75	55.25	43.7	26.3	79.5	101.5
55.25	55.25	0.0	0.0	75.0	0.0
57.75	55.25	−44.2	26.5	79.5	−102.3
60.25	62.75	44.8	26.8	79.7	103.0
62.75	62.75	0.0	0.0	75.0	0.0
65.25	62.75	−45.0	26.9	79.7	−103.1
67.75	70.25	45.0	26.8	79.7	102.4
70.25	70.25	0.0	0.0	75.0	0.0
72.75	70.25	−44.8	26.7	79.6	−101.7
75.25	77.75	44.3	26.4	79.5	99.9
77.75	77.75	0.0	0.0	75.0	0.0
80.25	77.75	−44.0	26.2	79.4	−98.7
82.75	85.25	43.1	25.6	79.2	95.9
85.25	85.25	0.0	0.0	75.0	0.0
87.75	85.25	−42.6	25.3	79.1	−94.4
90.25	92.75	41.5	24.6	78.9	91.1
92.75	92.75	0.0	0.0	75.0	0.0
95.25	92.75	−40.8	24.2	78.8	−89.3
97.75	100.25	39.5	23.4	78.6	85.7
100.25	100.25	0.0	0.0	75.0	0.0
102.75	100.25	−38.8	23.0	78.5	−83.8



Figure 1. Contour plots of the systematic component of panel error for two adjacent panels, taken from the scheme used for Table 2. The top panel, which is centered at $r = 67.75$ m, is cut to match the surface at $r_{\rm design} = 70.25$ m; the bottom panel, centered at $r = 65.25$ m, is cut appropriately for $r_{\rm design} = 62.75$ m. The contour interval is 5 μm; the zero- and positive contours are solid, and the negative ones are dashed. The error at the panel centers is ± 45 μm, owing to the offsets. The maximum errors occur at the panel corners. There is a discontinuity along the seam between the two panels; the magnitude of this discontinuity ranges between ~ 140 μm and ~ 205 μm. There would be no discontinuity along the left- and right-hand seams (except at corners). Along the seams between the panels that would be located immediately above and below the panels shown, the magnitude of the discontinuity would range between ~ 70 μm and ~ 100 μm, because those panels would have no (systematic) error.

Table 3. Twenty-Mold Calculation (2.5 m × 2 m Panels)

Panel Location, r (meters)	Design Radius (meters)	Offset (microns)	R.m.s., σ (microns)	$\sqrt{(75\mu\text{m})^2 + \sigma^2}$ (microns)	Max. |Error| (microns)
5.25	6.50	3.8	2.2	75.0	9.3
7.75	6.50	−4.6	2.7	75.0	−11.1
10.25	11.50	6.7	4.0	75.1	16.3
12.75	11.50	−7.4	4.4	75.1	−18.0
15.25	16.50	9.5	5.7	75.2	22.9
17.75	16.50	−10.2	6.1	75.2	−24.4
20.25	21.50	12.0	7.2	75.3	28.9
22.75	21.50	−12.6	7.6	75.4	−30.3
25.25	26.50	14.4	8.6	75.5	34.3
27.75	26.50	−14.9	9.0	75.5	−35.6
30.25	31.50	16.4	9.9	75.6	39.0
32.75	31.50	−16.8	10.1	75.7	−40.1
35.25	36.50	18.1	10.9	75.8	42.9
37.75	36.50	−18.5	11.1	75.8	−43.8
40.25	41.50	19.5	11.8	75.9	46.1
42.75	41.50	−19.8	11.9	75.9	−46.7
45.25	46.50	20.7	12.4	76.0	48.4
47.75	46.50	−20.9	12.6	76.0	−48.9
50.25	51.50	21.5	12.9	76.1	50.1
52.75	51.50	−21.6	13.0	76.1	−50.4
55.25	56.50	22.0	13.2	76.2	51.1
57.75	56.50	−22.1	13.3	76.2	−51.2
60.25	61.50	22.4	13.4	76.2	51.5
62.75	61.50	−22.4	13.4	76.2	−51.5
65.25	66.50	22.5	13.4	76.2	51.5
67.75	66.50	−22.5	13.4	76.2	−51.4
70.25	71.50	22.4	13.4	76.2	51.0
72.75	71.50	−22.4	13.3	76.2	−50.8
75.25	76.50	22.2	13.2	76.2	50.1
77.75	76.50	−22.1	13.2	76.1	−49.8
80.25	81.50	21.8	13.0	76.1	48.9
82.75	81.50	−21.7	12.9	76.1	−48.5
85.25	86.50	21.4	12.7	76.1	47.4
87.75	86.50	−21.2	12.6	76.1	−47.0
90.25	91.50	20.8	12.3	76.0	45.8
92.75	91.50	−20.7	12.2	76.0	−45.3
95.25	96.50	20.2	12.0	75.9	44.0
97.75	96.50	−20.0	11.9	75.9	−43.5
100.25	101.50	19.5	11.6	75.9	42.1
102.75	101.50	−19.3	11.5	75.9	−41.7

```
      program patchcalc4
c  This program calculates the r.m.s. difference (and minimum
c  and maximum mismatch) between the nominal design paraboloid
c  of the GBT and the nominal figure of its surface panels. In so doing,
c  it assumes that a panel is molded to be a close fit at some position
c  on the surface, specified by its radial distance from the vertex,
c  but then that that panel is used at other radial distances besides
c  the design radius.
      implicit real*8 (a-h,o-z)
      common rcent,rdesign,c,offset,rpsi,reta
      external fsq,g,h
c
c  Main parameters:
c
c     c: focal length of the design paraboloid, in meters, (fixed at c=60).
c     rcent: radial distance (sqrt(x^2+y^2)) of panel center; i.e.,
c            something in the range between 4 and 104 meters.
c     rdesign: radial distance that panel is cut (designed) to match;
c              i.e., something in the range between 4 and 104 meters.
      c=60d0
      pi=4d0*atan(1d0)
c  Read rmax:
      print *,'Type length,width'
      accept *,rpsi,reta
      rpsi=rpsi/2
      reta=reta/2
c  Read panel center location and place that panel is designed for:
1     print *,'Type rdesign,rcent'
      accept *,rdesign,rcent

      offset=0d0
      e1=f(rpsi,0d0)
      e2=f(rpsi,reta)
      e3=f(rpsi,-reta)
      e4=f(0d0,reta)
      e5=f(0d0,-reta)
      e6=f(-rpsi,0d0)
      e7=f(-rpsi,reta)
      e8=f(-rpsi,-reta)
      emin=min(e1,e2,e3,e4,e5,e6,e7,e8)
      emax=max(e1,e2,e3,e4,e5,e6,e7,e8)
      print *,'emax,emin=',emax*1d6,emin*1d6

      offset=.25d0*(emax+emin)
      print *,'offset=',offset*1d6
      emin=emin-offset
      emax=emax-offset
      print *,'emax,emin(offset)=',emax*1d6,emin*1d6

      errabs=1d-8
      errrel=1d-6
      irule=2
c  Integrate the square of the difference in heights above the two
c  tangent planes, subtracting off an offset of one-half the average
c  of the extreme differences on the edge of the patch.
      call dtwodq(fsq,-reta,reta,g,h,errabs,errrel,irule,result,err)
      print *,result,err
      result2=4*rpsi*reta
c  Calculate and print out the r.m.s. error:
      sigma=sqrt(result/result2)*1d6
```

```
      print *,'rms error (in microns)=',sigma
c  Repeat the calculation for another pair, (rdesign,rcent):
      go to 1
      end



      double precision function f(psi,eta)
c  Tangent planes are located at two distinct radii, rcent and rdesign,
c  from the vertex of the paraboloid.  For points on the paraboloid at
c  the same tangent plane coordinates above the two planes, this function
c  subroutine calculates the difference in heights (possibly adding an
c  offset, if the variable 'offset' is nonzero).
c  The psi-axis runs in the direction of positive r at the tangent point,
c  and the eta-axis in the direction of positive azimuth, phi.
      implicit real*8 (a-h,o-z)
      common rcent,rdesign,c,offset,rpsi,reta
      f=t(rcent,c,psi,eta)-t(rdesign,c,psi,eta)-offset
      return

      entry fsq
c  fsq is the square of f.
      fsq=(t(rcent,c,psi,eta)-t(rdesign,c,psi,eta)-offset)**2
      end

      double precision function t(r,c,psi,eta)
c  Given tangent plane coordinates (psi,eta) of a point on the
c  paraboloid, this function subroutine calculates the height of
c  that point above the tangent plane.  The point of tangency,
c  (psi,eta)=(0,0), is centered at a distance r=sqrt(x**2+y**2)
c  away from the vertex.  The height above the tangent plane is
c  measured along the direction of the surface normal at psi=eta=0.
      implicit real*8 (a-h,o-z)
      a=sqrt(r**2+4d0*c**2)
      t = (-a*sqrt(4d0*a*c*psi*r+(4d0*c**2-a**2)*eta**2+a**4)+
     1       a**3+2d0*c*psi*r)/r**2
      return
      end

      double precision function g(x)
c  Function subroutines g and h simply set the limits of integration
c  for the inner integral calculated by the two-dimensional numerical
c  quadrature routine dtwodq.
      implicit real*8 (a-h,o-z)
      common rcent,rdesign,c,offset,rpsi,reta
      g=-rpsi
      return
      end
      double precision function h(x)
      implicit real*8 (a-h,o-z)
      common rcent,rdesign,c,offset,rpsi,reta
      h=rpsi
      return
      end
```